

```

//*****
// This macro converts Epson printer code pairs
// from RMS-III to WP-6.0 character attribute pairs.
//*****

//Only run from main edit screen
IF(NOT ?EditScreen)
    PROMPT("Error: Macro can't be run in this substructure")
    PAUSE
ENDIF

SAVESTATE PERSISTALL
AutoCodePlacement(OFF!) WP51CursorMovement(ON!) VARERRCHK(OFF!)

Ttl="RMS-III Conversion" //Title

CANCEL(ON!)
ONCANCEL CALL(End2@)

DLGCREATE(DlgRet;Ttl;32;;;56;12)
    DLGCONTROL(CtrlLabel!;;" WordPerf 6.0 DOS macro to convert Epson-type";;StyNoBox!;;;46;1)
    DLGCONTROL(CtrlLabel!;;" attribute/print codes to WP attributes. These";;StyNoBox!;;;46;1)
    DLGCONTROL(CtrlLabel!;;" are the codes that when brought into WP appear ";;StyNoBox!;;;46;1)
    DLGCONTROL(CtrlLabel!;;" as ←G, ←-1, ←4, etc. (← is {ESC}). The process ";;StyNoBox!;;;46;1)
    DLGCONTROL(CtrlLabel!;;" makes three loops through the file, one each for";;StyNoBox!;;;46;1)
    DLGCONTROL(CtrlLabel!;;" bold, underline, and italic. The screen will not";;StyNoBox!;;;46;1)
    DLGCONTROL(CtrlLabel!;;" display the results until process is complete.";;StyNoBox!;;;46;1)
DLGEND

IF(DlgRet=-1) //If user canceled Dlg
    GO(End2@)
ENDIF

PROMPT("Processing, please wait...")

DISPLAY(OFF!)

PosDocTop

NOTFOUND(ON!) ONNOTFOUND(fixbold) LABEL(loop1)

SearchString("←G")
SearchNext()
DeleteCharPrevious
DeleteCharPrevious
BlockOn(CharMode!)
SearchString("←H")
SearchNext()
AttributeAppearanceOn(Bold!)
BlockOff
PosCharPrevious
DeleteCharPrevious
DeleteCharPrevious
//DeleteCharNext

GO(loop1)
LABEL(fixbold)
PosDocTop

NOTFOUND(ON!) ONNOTFOUND(fixund) LABEL(loop2)

```

```
SearchString("←-1")
SearchNext()
DeleteCharPrevious
DeleteCharPrevious
DeleteCharPrevious
BlockOn(CharMode!)
SearchString("←-0")
SearchNext()
AttributeAppearanceOn(Underline!)
BlockOff
PosCharPrevious
DeleteCharPrevious
DeleteCharPrevious
DeleteCharPrevious
//DeleteCharNext
```

```
GO(loop2)
LABEL(fixund)
PosDocTop
```

```
NOTFOUND(ON!) ONNOTFOUND(fixitalic) LABEL(loop3)
```

```
PosDocTop
SearchString("←4")
SearchNext()
DeleteCharPrevious
DeleteCharPrevious
BlockOn(CharMode!)
SearchString("←5")
SearchNext()
AttributeAppearanceOn(Italics!)
BlockOff
PosCharPrevious
DeleteCharPrevious
DeleteCharPrevious
```

```
GO(loop3)
LABEL(fixitalic)
PosDocTop
```

```
DISPLAY(ON!)
```

```
DLGCREATE(x;"",DlgNoCancel!;;;30;7)
  DLGCONTROL(CtrlLabel!;;" Conversion complete.";StyNoBox!;;;20;1)
  DLGCONTROL(CtrlLabel!;;" Return to document.";StyNoBox!;;;20;1)
DLGEND
```

```
LABEL(end2@)
```

```
QUIT
```